Oscillators

Unforced Oscillators

```
> restart:with(plots):with(DEtools):
```

We want to consider unforced oscillators of the form

$$my'' + \mathbf{by}' + ky = 0,$$

where m is the mass, b is the damping constant, and k is the spring constant. In what follows, we will assume $m = 1$, $k = \beta^2$, and $b \geq 0$. The characteristic equation is

$$r^2 + br + k = 0,$$

so

$$r = \frac{-b \pm \sqrt{b^2 - 4\cdot\beta^2}}{2}.$$

Undamped Oscillators

When $b = 0$, we have $b^2 - 4\cdot\beta^2<0$, so we have complex roots and the real part is $-\frac{b}{2}$, which is always negative. We say we have an undamped oscillator. An example of this is the horizontal mass-spring system discussed at the beginning of Chapter 4. The differential equation is now $y'' + \beta^2 y = 0$, so the characteritic equation will have conjugate complex roots iβ and -iβ. For example, let's look at y'' + 4y = 0 with initial conditions y(0)=5 and y'(0)=1.

```
> deq1:=(D@@2)(x)(t)+4*x(t)=0;
```

$$deq1 := D^{(2)}(x)(t) + 4\,x(t) = 0$$

```
> IC:=x(0)=5,D(x)(0)=1;
```

$$IC := x(0) = 5, D(x)(0) = 1$$

```
> soln1:=dsolve({deq1,IC},x(t));
```

$$soln1 := x(t) = \frac{1}{2}\sin(2\,t) + 5\cos(2\,t)$$

We plot the solution for t positive and less than 8.

```
> p1:=plot(rhs(soln1),t=0..8,color=blue):
  display(p1);
```



You can see why we call it an oscillator.

Underdamped Motion

Now lets put in a **damping** term **by**$'$ where $0 < b$. This corresponds to a frictional or damping force. Suppose $b < 2\beta$, or for our example, $b < 4$. Then the characteristic equation will still have complex congugate roots. For instance, choose $b = 1$.

```
> deq2:=(D@@2)(x)(t)+D(x)(t)+4*x(t)=0;
```

$$deq2 := D^{(2)}(x)(t) + D(x)(t) + 4\,x(t) = 0$$

```
> soln2:=dsolve({deq2,IC},x(t));
```

$$soln2 := x(t) = \frac{7}{15}\sqrt{15}\,e^{-\frac{1}{2}t}\sin\left(\frac{1}{2}\sqrt{15}\,t\right) + 5\,e^{-\frac{1}{2}t}\cos\left(\frac{1}{2}\sqrt{15}\,t\right)$$

We plot both the undamped (blue) and damped (red) oscillators.

```
> p2:=plot(rhs(soln2),t=0..8,color=red):
  display(p1,p2);
```



We still have oscillation, but the damping effect is clear. With $b < 2\beta$, we say that the system is **underdamped** because there is not enough damping present (b is too small) to prevent the system from oscillating. Now choose $b = 2$, still giving complex roots for the characteristic equation.

```
> deq3:=(D@@2)(x)(t)+2*D(x)(t)+4*x(t)=0;
```

$$deq3 := D^{(2)}(x)(t) + 2\,D(x)(t) + 4\,x(t) = 0$$

```
> soln3:=dsolve({deq3,IC},x(t));
```

$$soln3 := x(t) = 2\sqrt{3}\,e^{-t}\sin\left(\sqrt{3}\,t\right) + 5\,e^{-t}\cos\left(\sqrt{3}\,t\right)$$

We plot the undamped (blue) and damped (red and green) oscillators.

```
> p3:=plot(rhs(soln3),t=0..8,color=green):
  display(p1,p2,p3);
```



We clearly have more damping with the higher value of *b*.

Critically damped motion.

Now let *b*=2 β=4. We will now get a single real repeated root for our characteristic equation.

```
> deq4:=(D@@2)(x)(t)+4*D(x)(t)+4*x(t)=0;
```

$$deq4 := D^{(2)}(x)(t) + 4\,D(x)(t) + 4\,x(t) = 0$$

```
> soln4:=dsolve({deq4,IC},x(t));
```

$$soln4 := x(t) = 5\,e^{-2\,t} + 11\,e^{-2\,t}\,t$$

We see our solution no longer has sin or cos in it, so we should not expect oscillation.

```
> p4:=plot(rhs(soln4),t=0..8,color=magenta):
  display(p1,p4);
```



Here we get neither a max or a min. With b=2β, we say the system is **critically damped**, since if *b* were any smaller, oscillation would occur.

Overdamped Motion

Now choose *b*>2 β. We take *b* = 6 here.

```
> deq5:=(D@@2)(x)(t)+16*D(x)(t)+4*x(t)=0;
```

$$deq5 := D^{(2)}(x)(t) + 16\,D(x)(t) + 4\,x(t) = 0$$

```
> soln5:=dsolve({deq5,IC},x(t));
```

$$soln5 := x(t) = \left(\frac{5}{2} + \frac{41}{60}\sqrt{15}\right) e^{2\left(-4+\sqrt{15}\right)t} + \left(\frac{5}{2} - \frac{41}{60}\sqrt{15}\right) e^{-2\left(4+\sqrt{15}\right)t}$$

Again we see our solution has no sin or cos in it, so we should not expect oscillation.

```
> p5:=plot(rhs(soln5),t=0..8,color=cyan):
  display(p1,p5);
```



Here again we get neither a max or a min. With $b>2\,\beta$, we say the system is **overdamped**. The graphs of the critically damped and overdamped systems will have at most one extreme point, either a max or a min, but may have neither as has happened here.

Forced Oscillators

```
> restart:with(plots):
```

We consider the vibrations of a spring-mass system when an external force is applied. We choose the interesting case of a periodic forcing function, and consider the following forced oscillator:

$$y'' + .3\,y' + 4.9\,y = \cos(4\,t)$$

with initial conditions

$$y(0) = 1 \text{ and } y'(0) = 1.$$

We first look at the unforced oscillator without the forcing cos term..

```
> unforced:=diff(y(t),t$2)+(3/10)*diff(y(t),t)+(49/10)*y(t)=0;
```

$$unforced := \frac{d^2}{dt^2}\,y(t) + \frac{3}{10}\,\frac{d}{dt}\,y(t) + \frac{49}{10}\,y(t) = 0$$

We solve the unforced oscillator.

```
> transient:=dsolve(unforced,y(t));
```

$$transient := y(t) = \_C1\, e^{-\frac{3}{20} t} \sin\left(\frac{1}{20}\sqrt{1951}\, t\right) + \_C2\, e^{-\frac{3}{20} t} \cos\left(\frac{1}{20}\sqrt{1951}\, t\right)$$

We refer to this solution of this underdamped unforced oscillator as the transient solution since it approaches 0 as t approaches infinity (i.e., it sort of disappears) because of the exponentials. Now we add in the oscillating forcing function.

```
> forced:=diff(y(t),t$2)+(3/10)*diff(y(t),t)+(49/10)*y(t)=2*cos(4*t);
```

$$forced := \frac{d^2}{dt^2} y(t) + \frac{3}{10} \frac{d}{dt} y(t) + \frac{49}{10} y(t) = 2\cos(4t)$$

We solve the forced oscillator.

```
> solution:=dsolve(forced,y(t));
```

$$solution := y(t) = e^{-\frac{3}{20} t} \sin\left(\frac{1}{20}\sqrt{1951}\, t\right) \_C2 + e^{-\frac{3}{20} t} \cos\left(\frac{1}{20}\sqrt{1951}\, t\right) \_C1 + \frac{16}{831}\sin(4t) - \frac{148}{831}\cos(4t)$$

The difference of the solutions of the forced and unforced oscillators is a particular solution. We also refer to this difference as the steady state solution.

```
> steadystate:=-148/831*cos(4*t)+16/831*sin(4*t);
```

$$steadystate := \frac{16}{831}\sin(4t) - \frac{148}{831}\cos(4t)$$

We check that this really is a particular solution.

```
> eval(subs(y(t)=steadystate,forced));
```

$$2\cos(4t) = 2\cos(4t)$$

The equality indicates that we have a particular solution. Now we solve the forced oscillator with the initial conditions.

```
> IC:=y(0)=1,D(y)(0)=1;
```

$$IC := y(0) = 1, D(y)(0) = 1$$

```
> solutionIC:=dsolve({forced,IC},y(t));
```

$$solutionIC := y(t) = \frac{18277}{1621281} e^{-\frac{3}{20} t} \sqrt{1951} \sin\left(\frac{1}{20}\sqrt{1951}\, t\right) + \frac{979}{831} e^{-\frac{3}{20} t} \cos\left(\frac{1}{20}\sqrt{1951}\, t\right) + \frac{16}{831}\sin(4t) - \frac{148}{831}\cos(4t)$$

We now plot the solution (blue) **along with the steady state (green).**

```
> p1:=plot(rhs(solutionIC),t=0..40,color=blue,numpoints=10000):
  p2:=plot(steadystate,t=0..40,color=green,numpoints=10000):
  display(p1,p2);
```



We see that as t increases to infinity and the transient solution goes to 0, the actual solution approaches the steady state solution.